Daten per Office Automation an Excel übergeben

 Teil 1
Versionen: Access 2010, 2007, 2003, 2002/XP und 2000

In Ihren Datenbanken müssen Sie regelmäßig Daten an Excel übergeben. Dazu können Sie die eingebauten Funktionen "TransferSpreadsheet" oder "OutputTo" nutzen. Diese Funktionen sind jedoch recht unflexibel, da unter anderem der Zielbereich nur eingeschränkt definiert und lediglich ein vorhandenes Tabellenblatt verwendet werden kann. Eine interessante Alternative ist die Übertragung per Office Automation, bei der Sie alle Optionen bis hin zum automatischen Anlegen neuer Tabellenblätter flexibel nutzen können. Dieser zweiteilige Tipp der Woche vermittelt Grundlagen und stellt fertige Lösungen vor.

Office Automation ist eine Technik, mit der Sie aus einer Office-Anwendung wie Access eine andere Office-Anwendung wie Excel fernsteuern können. Basis ist ein "Objektmodell", das Informationen über Daten und Funktionen der fernzusteuernden Anwendung bereitstellt. Um also Excel aus Access heraus fernsteuern zu können, muss im ersten Schritt das Objektmodell von Excel in Access eingebunden werden. Dazu öffnen Sie zunächst die Datenbank, aus der ein Zugriff auf Excel erfolgen soll. Dort wechseln Sie mit Alt+F11 in die VBA-Entwicklungsumgebung und rufen den Dialog Extras-Verweise auf. Lokalisieren Sie den Eintrag "Microsoft Excel x.x Object Library", aktivieren Sie das Kontrollkästchen vor dem Eintrag. "x.x" steht für die jeweilige Excel-Version. Weiterhin wird für die folgenden Funktionen eine Referenz auf "Microsoft DAO 3.6 Object Library" benötigt, für das Sie ebenfalls das Kontrollkästchen aktivieren. Ein Klick auf Ok übernimmt beide Einbindungen.

Um nun auf Excel zuzugreifen und beispielsweise den Inhalt einer Access-Tabelle auf ein Excel-Tabellenblatt zu übertragen, muss zunächst eine Verbindung zu Excel aufgebaut werden:

Dim objExcel As Excel.Application
Dim ws As Excel.Worksheet
Dim wb As Excel.Workbook

On Error Resume Next
Set objExcel = CreateObject("Excel.Application")
If Err <> 0 Then
  Beep
  MsgBox "Verbindung zu Excel kann nicht " & _
         "aufgebaut werden: " & Err.Description
  Exit Sub
End If

Die Funktion "CreateObject()" startet eine neue Instanz von Excel. Ein möglicherweise auftretender Fehler wird per "MsgBox" gemeldet. Andernfalls halten wir die gelieferte Referenz auf das Objektmodell von Excel in der Objektvariablen "objExcel" fest. Darüber können nun weitere Excel-Befehle abgesetzt werden. Soll beispielsweise eine existierende Arbeitsmappe verwendet werden, öffnen Sie sie mit der folgenden Anweisung:

Set wb = objExcel.Workbooks.Add("C\Test\Test.xls")

Die Auflistung "Workbooks" verwaltet alle geöffneten Arbeitsmappen, zu denen mit "Add" weitere hinzugefügt werden können. Als Parameter übergeben Sie dabei den entsprechenden Pfad und Dateinamen. Das Ergebnis halten wir in einer Objektvariablen "wb" fest, über die dann der weitere Zugriff auf die Arbeitsmappe erfolgt.

Möchten Sie hingegen mit einer neuen Arbeitsmappe beginnen, können Sie die normalerweise beim Start von Excel automatisch angelegte Arbeitsmappe nutzen. Da der Anwender diesen Automatismus jedoch in Excel abgestellt haben kann, muss eine entsprechende Prüfung erfolgen:

If objExcel.Workbooks.Count = 0 Then
  Set wb = objExcel.Workbooks.Add()
End If

"Workbooks.Count" gibt Auskunft, wie viele Arbeitsmappen derzeit geöffnet sind. Ist die Anzahl "0", also ist keine Arbeitsmappe geöffnet, wird mit "Add" eine neue und anschließend über "wb" referenzierbare Arbeitsmappe angelegt.

"wb" referenziert nun die aktuelle Arbeitsmappe, in der Sie beispielsweise wie folgt ein bestimmtes Tabellenblatt adressieren können:

Set ws = wb.Sheets("NameDesTabellenblattes")

In der Auflistung "Sheets" werden alle Tabellenblätter der aktuellen Arbeitsmappe verwaltet und können über den Namen eines Tabellenblattes oder über einen Index von "1" bis "Sheets.Count" adressiert werden. Anschließend referenziert "ws" das gewünschte Tabellenblatt in der Arbeitsmappe. Sollen die Daten auf ein neues Tabellenblatt übertragen werden, legen Sie es wie folgt an:

Set ws = wb.Sheets.Add()
ws.Name = "NameDesTabellenblattes"

"Sheets.Add" fügt ein neues Tabellenblatt mit einem Namen wie "Tabelle4" zur aktuellen Arbeitsmappe hinzu. Mit "ws.Name" wird dieser Name wie gewünscht geändert.

Damit wissen Sie nun, wie man eine Verbindung zu Excel aufbaut, eine vorhandene oder neue Arbeitsmappe nutzt und hier ein vorhandenes oder neues Tabellenblatt adressiert. Im zweiten Teil zeigen wir, wie Sie mit Hilfe einer wenig bekannten Excel-Methode komplette Access-Tabellen oder Abfrageergebnisse mit einer einzigen Anweisung in das Excel-Tabellenblatt übertragen.

Im ersten Teil dieses Tipps der Woche haben wir gezeigt, wie man per Office Automation eine Verbindung zu Excel aufbaut, eine vorhandene oder neue Arbeitsmappe nutzt und hier ein vorhandenes oder neues Tabellenblatt adressiert. Im zweiten Teil geht es nun darum, mit Hilfe einer wenig bekannten Excel-Methode komplette Access-Tabellen oder Abfrageergebnisse mit einer einzigen Anweisung in ein Excel-Tabellenblatt zu übertragen.

Die Rede ist von der Methode "CopyFromRecordset", die ab einer bestimmten Zelle alle Datensätze eines Recordsets einfügt. Dazu muss zunächst ein Recordset initialisiert werden:

Dim db As DAO.Database
Dim rs As DAO.Recordset

Set db = CurrentDb()
Set rs = db.OpenRecordset("Artikel")

Diese Anweisungen öffnen eine Tabelle "Artikel" als Recordset. Um das Ergebnis einer Abfrage an Excel zu übergeben, können Sie anstelle eines Tabellennamens auch den Namen der betreffenden Abfrage oder eine SQL-Anweisung der Form "select * from Artikel where Auslaufartikel= Yes order by Lagerbestand" angeben. Ist das Recordset initialisiert, übertragen Sie die Datensätze mit folgender Anweisung in das per "ws" adressierte Tabellenblatt:

ws.Range("C10").CopyFromRecordset rs

Mit Range() legen Sie die obere linke Zelle fest, ab der die Datensätze untereinander eingefügt werden sollen - "C10" adressiert hier in der 3. Spalte die 10. Zeile, soll einfach das Tabellenblatt gefüllt werden, geben Sie "A1" für die Startzelle an. Bitte beachten Sie, dass "CopyFromRecordset" vorhandene Zellinhalte ohne Rückfrage überschreibt! Dadurch besteht einerseits die Gefahr, dass vorhandene Daten verloren gehen, andererseits gestalten sich dadurch Aktualisierungen sehr einfach.

Wenn das Recordset in das Tabellenblatt übertragen ist, muss schließlich noch die Arbeitsmappe angezeigt werden. Das erledigt die folgende Anweisung:

objExcel.Visible = True

Hintergrund: Per Office Automation initiierte Instanzen werden normalerweise unsichtbar gestartet, um den Anwender nicht durch unnötig angezeigte Fenster zu verwirren. Mit "Visible = True" wird die Instanz sichtbar gemacht, damit der Anwender die Daten bearbeiten oder kontrollieren kann.

Mit den bisher vermittelten Grundlagen lässt sich nun eine universell nutzbare Funktion für die Übertragung von Access-Daten an Excel aufbauen:

Function DataToExcel( _
         strDataSource As String, _
         Optional strWorkbook As String = "", _
         Optional strSheet As String = "", _
         Optional strNewSheet As String = "", _
         Optional strStartAt As String = "A1") As Long

  'Access, DAO
  Dim db As DAO.Database
  Dim rs As DAO.Recordset
  'Excel
  Dim objExcel As Excel.Application
  Dim ws As Excel.Worksheet
  Dim wb As Excel.Workbook

  On Error Resume Next
  Set objExcel = CreateObject("Excel.Application")
  If Err <> 0 Then
    Beep
    MsgBox "Verbindung zu Excel kann nicht " & _
           "aufgebaut werden: " & Err.Description
    Exit Function
  End If

  If strWorkbook = "" Then
    Set wb = objExcel.Workbooks.Add()
  Else
    Set wb = objExcel.Workbooks.Open(strWorkbook)
  End If

  If strSheet = "" Then
    Set ws = wb.Sheets.Add()
    If strNewSheet <> "" Then
      ws.Name = strNewSheet
    End If
  Else
    Set ws = wb.Sheets(strSheet)
  End If

  Set db = CurrentDb()
  Set rs = db.OpenRecordset(strDataSource)

  ws.Range(strStartAt).CopyFromRecordset rs
  DataToExcel = rs.RecordCount

  objExcel.Visible = True

  Set rs = Nothing
  Set db = Nothing

End Function

Der erste Parameter spezifiziert die Datenquelle; hier geben Sie also den Namen einer Tabelle oder Abfrage oder eine SQL-Anweisung "select .. from" an. Alle weiteren Parameter sind optional, so dass Sie beispielsweise per

DataToExcel "Kunden"

die Datensätze aus einer Tabelle "Kunden" einfach in eine neue Arbeitsmappe auf ein neues Tabellenblatt übertragen und Arbeitsmappe und Tabellenblatt anschließend nach belieben benennen können. Um beispielsweise Artikeldaten in eine vorhandene Arbeitsmappe in ein neues Tabellenblatt "Artikelübersicht" zu übertragen oder dort zu aktualisieren, nutzen Sie folgende Anweisung:

DataToExcel "Artikel", _
            "Z:\Test\Artikel.xls", _
            "Artikelübersicht"

Ein weiteres Beispiel: Die folgenden Anweisungen übertragen die Bestellungen des aktuellen Tages, selektiert per Abfrage "Bestellungen heute", in eine vorhandene Arbeitsmappe auf ein neues mit Datumsstempel versehenes Tabellenblatt:

DataToExcel "Bestellungen heute", _
            "Z:\Test\Bestellungen.xls", _
            "", _
            "Bestellungen_" & Format$(Now,"dd.mm.yyyy")
